

# Fourier-Reihen Rechteckschwingung

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 4, Sept 07 Update 14.05.08

Web: <http://haftendorn.uni-lueneburg.de> [www.mathematik-verstehen.de](http://www.mathematik-verstehen.de)

#####


Dateiname fourier-reihen-rechteck.mn

Angabe der Funktion fH in der Hauptperiode.

Definition von f als periodische Funktion.

Achtung: eine volle Periode **im Positiven** als Hauptfunktion beschreiben!

```
Ta:=2.5:Ti1:=3:Ti2:=7: Te:=7.5:
fH:=t--> piecewise([Ta<=t and t<Ti1,0],
                  [Ti1<=t and t<=Ti2,3],[Ti2<=t and t<=Te,0]) :
f:=t->fH(frac((t-(Te-Ta)/2)/(Te-Ta))*(Te-Ta)+(Te-Ta)/2):
plotfunc2d(fH(t), f(t), t=-5..10,
           Scaling=Constrained, LegendVisible=FALSE)
```



Definitionen von Feldern für die Koeffizienten, übliche Festlegungen T Periode,  $\omega = \text{omega}$  Kreisfrequenz.

```
n:=10:
a:=array(0..10):      b:=array(1..10):
T:=5:
om:=2*PI/T:
```

Das Verschiebungsglied kann oft auch elementargeometrisch bestimmt werden. Hier als Dreieck  $g=4, h=6 \rightarrow F=12$  und  $2/T=2/4=1/2$

```
a[0]:=2/T*int(fH(t), t=Ta..Te)
```

$$\frac{24}{5}$$

So kann man nur die anderen Koeffizienten berechnen.

```
a[1]:=Simplify(2/T*int(fH(t)*cos(1*om*t), t=Ta..Te));
b[1]:=2/T*int(fH(t)*sin(1*om*t), t=Ta..Te);
```

$$\frac{3 \cdot \sqrt{2} \cdot \sqrt{5 - \sqrt{5}}}{2 \cdot \pi}$$

1

0

0

Automatische Berechnung von n Schwingungen

```
for k from 1 to 10 do
    a[k]:=2/T*int(fH(t)*cos(k*om*t),t=Ta..Te);
    b[k]:=0: /*2/T*int(fH(t)*sin(k*om*t),t=Ta..Te);*/
end_for:
```

For-Schleifen haben keine Ausgaben. Ausgabe folgt. Vereinfachen muss extra gefordert werden.

```
[Simplify(a[k]),Simplify(b[k])] $ k=1..n
```

$$\left[ \frac{3 \cdot \sqrt{2} \cdot \sqrt{5 - \sqrt{5}}}{2 \cdot \pi}, 0 \right], \left[ -\frac{3 \cdot \sqrt{2 \cdot \sqrt{5} + 10}}{4 \cdot \pi}, 0 \right], \left[ \frac{\sqrt{2 \cdot \sqrt{5} + 10}}{2 \cdot \pi}, 0 \right], \left[ -\frac{3 \cdot \sqrt{2} \cdot \sqrt{5}}{8 \cdot \pi} \right]$$

Wenn man mit so einem Werkzeug arbeitet, braucht man über die Gesetzmäßigkeit in den Ausgaben nicht nachzudenken. Falls nötig, könnte man es schaffen.


Im Folgenden wird die Summe bis Ordnung n gebildet.


```
su:=(t,n)->a[0]/2+_plus((a[k]*cos(k*om*t)+b[k]*sin(k*om*t)) $ k=1..n);
```

$$(t, n) \rightarrow \frac{a_0}{2} + \text{\_plus}(a_k \cdot \cos(k \cdot \text{om} \cdot t) + b_k \cdot \sin(k \cdot \text{om} \cdot t) \text{ \$ } k = 1 \dots n)$$


Die Grundschiwingung und ihre Achse ist deutlich.

```
plotfunc2d(fH(t),su(t,n) $
n=1..10,a[0]/2,LegendVisible=FALSE)
```





`plotfunc2d(su(t,10),f(t), LegendVisible=FALSE)`




Die Reihen mit 10 Doppeltermen ist von der gegebenen Funktion noch deutlich zu unterscheiden.

#####

Vergleich mit der Handrechnung, Bronsteinformel und meiner Folie

$$\begin{aligned}
 a_k &= \frac{2}{T} \int_{-a}^a f(t) \cos(k\omega t) dt \\
 &= \frac{2}{T} \cdot c \int_{-a}^a \cos(k\omega t) dt + 0 \\
 &= \frac{2}{T} c \frac{1}{k\omega} [\sin(k\omega t)]_{-a}^a = \frac{2c}{T k\omega} (\sin(k\omega a) - \sin(k\omega(-a))) \\
 &= \frac{2}{T} c \frac{1}{k\omega} \cdot 2 \sin(k\omega a) = \frac{2c}{k\pi} \sin(k\omega a) \quad \text{mit } \omega = \frac{2\pi}{T}
 \end{aligned}$$



Speziell  $a=2$   $c=3$

$$a_k = \frac{6}{k\pi} \sin\left(\frac{4k\pi}{T}\right) = \frac{6}{2\pi} \sin\left(\frac{4k\pi}{5}\right)$$

$$\text{Fourier term}(k) \stackrel{T=5}{=} = \frac{6}{2\pi} \sin\left(\frac{4k\pi}{5}\right) \cdot \cos\left(\frac{2\pi k t}{5}\right)$$

```

for k from 1 to 10 do
  c[k] := 6/k*PI*sin(4*k*PI/5);
  d[k] := 0;
end_for:

```

```

simplify(c[k]) $ k=1..5

```

$$\frac{3 \cdot \pi \cdot \sqrt{2} \cdot \sqrt{5 - \sqrt{5}}}{2}, -\frac{3 \cdot \pi \cdot \sqrt{2} \cdot \sqrt{\sqrt{5} + 5}}{4}, \frac{\pi \cdot \sqrt{2} \cdot \sqrt{\sqrt{5} + 5}}{2}, -\frac{3 \cdot \pi \cdot \sqrt{2} \cdot \sqrt{5 - \sqrt{5}}}{8}$$

Die Wurzeln erscheinen, weil sie 5-Teilung des Vollwinkels konstruiert werden kann.

Mit Siebenteln ginge es nicht.

```

for k from 1 to 10 do
  c[k] := 6/k*PI*sin(4*k*PI/7);
  d[k] := 0;
end_for:

```

```

simplify(c[k]) $ k=1..2

```

$$6 \cdot \pi \cdot \sin\left(\frac{3 \cdot \pi}{7}\right), -3 \cdot \pi \cdot \sin\left(\frac{\pi}{7}\right)$$